  PGM        PARM(&QPGM &QSRCFILE &SRCMBR &PGMTEXT +

                  &OPTIONS &REPLACE &AUT)

             DCL        VAR(&QPGM) TYPE(*CHAR) LEN(20)

             DCL        VAR(&QSRCFILE) TYPE(*CHAR) LEN(20)

             DCL        VAR(&SRCMBR) TYPE(*CHAR) LEN(10)

             DCL        VAR(&PGMTEXT) TYPE(*CHAR) LEN(50)

             DCL        VAR(&OPTIONS) TYPE(*CHAR) LEN(165)

             DCL        VAR(&REPLACE) TYPE(*LGL) LEN(1)

             DCL        VAR(&AUT) TYPE(*CHAR) LEN(10)

             DCL        VAR(&SRCFILE) TYPE(*CHAR) LEN(10)

             DCL        VAR(&SRCFILELIB) TYPE(*CHAR) LEN(10)

             DCL        VAR(&SRCCHGDATE) TYPE(*CHAR) LEN(13)

             DCL        VAR(&SRCMBRTXT) TYPE(*CHAR) LEN(50)

       /*  Split-up qualified names (only the ones necessary). */

             CHGVAR     VAR(&SRCFILE) VALUE(%SST(&QSRCFILE 1 10))

             CHGVAR     VAR(&SRCFILELIB) VALUE(%SST(&QSRCFILE 11 10))

       /*  Set/retrieve source info. */

             IF         COND(&SRCMBR = '*PGM') THEN(CHGVAR +

                          VAR(&SRCMBR) VALUE(&QPGM))

             RTVMBRD    FILE(&SRCFILELIB/&SRCFILE) MBR(&SRCMBR) +

                          RTNLIB(&SRCFILELIB) +

                          SRCCHGDATE(&SRCCHGDATE) TEXT(&SRCMBRTXT)

             IF         COND(&PGMTEXT = '*SRCMBRTXT') THEN(CHGVAR +

                          VAR(&PGMTEXT) VALUE(&SRCMBRTXT))

             CHGVAR     VAR(&QSRCFILE) VALUE(&SRCFILE *CAT &SRCFILELIB)

       /*  Set public authority. */

             IF         COND(&AUT = '*LIBCRTAUT') THEN(DO)

               RTVLIBD    LIB(%SST(&QPGM 11 10)) CRTAUT(&AUT)

               IF         COND(&AUT = '*SYSVAL') THEN(RTVSYSVAL +

                            SYSVAL(QCRTAUT) RTNVAR(&AUT))

             ENDDO

       /*  Override files and call CRTMIPGM1. */

             OVRPRTF    FILE(QSYSPRT) SPLFNAME(%SST(&QPGM 1 10))

             OVRDBF     FILE(QMISRC) TOFILE(&SRCFILELIB/&SRCFILE) +

                          MBR(&SRCMBR)

             CALL       PGM(CRTMIPGM1) PARM(&QPGM &PGMTEXT &QSRCFILE +

                          &SRCMBR &SRCCHGDATE 'QSYSPRT   *LIBL     ' +

                          &AUT &OPTIONS &REPLACE)

             DLTOVR     FILE(*ALL)

             SNDPGMMSG  MSGID(CPC0815) MSGF(QCPFMSG) MSGDTA(&QPGM) +

                          MSGTYPE(*COMP)

  ENDPGM




